

# Solutions to Week 3 Activities

---

## Solutions to Activities

### 1. Displaying orion.order\_fact with the PRINT Procedure

```
proc print data=orion.order_fact noobs;  
  where Total_Retail_Price>500;  
  id Customer_ID;  
  var Order_ID Order_Type Quantity Total_Retail_Price;  
  sum Total_Retail_Price;  
run;
```

- a. Run the program and view the output.
- b. Add a SUM statement and verify the resulting sum.
- c. What do you notice about the Obs column? **The numbers are not sequential. The original observation numbers are displayed.**  
Did the sum of **Total\_Retail Price** change to reflect only the subset? **Yes**
- d. If the Obs column is suppressed, how can you verify the number of observations in the results? **Check the log.**
- e. When the ID statement was added, how did the output change? **Customer\_ID is the leftmost column and is displayed on each line for an observation.**
- f. When the VAR statement is added, what do you notice about **Customer\_ID**? **There are two Customer\_ID columns. The first column is the ID field, and a second one is included because Customer\_ID is listed in the VAR statement.**
- g. Remove the duplicate column by removing **Customer\_ID** from the VAR statement.

### 2. Displaying orion.customer\_dim with the PRINT Procedure

```
proc print data=orion.customer_dim noobs;  
  where Customer_Age between 30 and 40;  
  id Customer_ID;  
  var Customer_Name Customer_Age Customer_Type;  
run;
```

### 3. Producing a Default Listing Report of orion.order\_fact (SAS Windowing Environment)

```
options ls=max;  
  
proc print data=orion.order_fact;  
run;  
  
options ls=96;  
  
proc print data=orion.order_fact headings=v;  
run;
```

- a. Submit a simple PROC PRINT step to produce a default listing report.

- b. What are the minimum and maximum values for the LINESIZE= option? **The minimum value for LINESIZE= is 64 and the maximum size is MAX.**

When you are finished, use the following statement to reset the line size to 96:

```
options ls=96;
```

- c. How do you specify vertical headings? **HEADINGS=V forces all column headings to be displayed vertically.**

How do you specify horizontal headings? **HEADINGS=H forces all column headings to be displayed horizontally.**

#### 4. Producing a Default Listing Report of orion.product\_dim (SAS Windowing Environment)

```
proc print data=orion.product_dim width=uniform;  
run;
```

- a. Submit a simple PROC PRINT step.
- b. Add the WIDTH=UNIFORM option. How are the results different? **Each column has the same column width on each page.**
- c. Why might the procedure run more slowly with this option? **With this option, PROC PRINT must read through the entire data set twice.**
- d. How can you save computer resources and still display columns consistently across pages? **Use a format on every column to explicitly specify a field width so that PROC PRINT reads the data only once.**

#### 5. Sorting orion.employee\_payroll and Displaying the New Data Set

```
proc sort data=orion.employee_payroll out=work.sort_salary;  
  by Salary;  
run;  
  
proc print data=work.sort_salary;  
run;
```

#### 6. Sorting orion.employee\_payroll and Displaying Grouped Observations

```
proc sort data=orion.employee_payroll out=work.sort_salary2;  
  by Employee_Gender descending Salary;  
run;  
  
proc print data=work.sort_salary2;  
  by Employee_Gender;  
run;
```

#### 7. Sorting orion.employee\_payroll and Displaying a Subset of the New Data Set

```
proc sort data=orion.employee_payroll out=work.sort_sal;
```

```

    by Employee_Gender descending Salary;
run;

proc print data=work.sort_sal noobs;
    by Employee_Gender;
    sum Salary;
    where Employee_Term_Date is missing and Salary>65000;
    var Employee_ID Salary Marital_Status;
run;

```

## 8. Retaining the First Observation of Each BY Group

```

proc sort data=orion.orders out=work.custorders nodupkey
    dupout=work.duplicates;
    by Customer_ID;
run;

title 'Unique Customers';
proc print data=work.custorders;
run;

title 'Duplicate Customer Observations';
proc print data=work.duplicates;
run;
title;

```

## 9. Displaying Titles and Footnotes in a Detail Report

```

title1 'Australian Sales Employees';
title2 'Senior Sales Representatives';
footnote1 'Job_Title: Sales Rep. IV';

proc print data=orion.sales;
    where Country='AU' and Job_Title contains 'Rep. IV';
    var Employee_ID First_Name Last_Name Gender Salary;
run;
title;
footnote;

```

## 10. Displaying Column Headings in a Detail Report

a.

```

title 'Entry-level Sales Representatives';
footnote 'Job_Title: Sales Rep. I';

proc print data=orion.sales noobs label;
  where Country='US' and Job_Title='Sales Rep. I';
  var Employee_ID First_Name Last_Name Gender Salary;
  label Employee_ID="Employee ID"
        First_Name="First Name"
        Last_Name="Last Name"
        Salary="Annual Salary";

run;

title;
footnote;

```

b.

```

title 'Entry-level Sales Representatives';
footnote 'Job_Title: Sales Rep. I';

proc print data=orion.sales noobs split=' ';
  where Country='US' and Job_Title='Sales Rep. I';
  var Employee_ID First_Name Last_Name Gender Salary;
  label Employee_ID="Employee ID"
        First_Name="First Name"
        Last_Name="Last Name"
        Salary="Annual Salary";

run;

title;
footnote;

```

## 11. Writing an Enhanced Detail Report

```

proc sort data=orion.employee_addresses out=work.address;
  where Country='US';
  by State City Employee_Name;
run;

title "US Employees by State";
proc print data=work.address noobs split=' ';
  var Employee_ID Employee_Name City Postal_Code;
  label Employee_ID='Employee ID'
        Employee_Name='Name'
        Postal_Code='Zip Code';
  by State;
run;

```